Función para mostrar datos
Un consejero experto normal (EA) que se usa en la práctica, es mucho más potente que un EA simple, porque proporciona información de alta calidad.  

El mercado siempre está cambiando y a todo momento suceden cosas. Para que el operador tome decisiones acertadas, es crucial que esté completamente informado  sobre la situación del mercado. Por esta razón, un EA debe tener una función que informe al operador. Esta función debe estar destinada a informar al usuario sobre un determinado conjunto de hechos y procesos. 
En general, un EA simple cumple esta labor utilizando la función Comment() que coloca un texto informativo en la parte superior izquierda de la grafica. Este método para mostrar información no es muy cómodo, puesto que en algunas ocasiones, el texto  puede tapar la grafica del precio. Este método se puede utilizar tan solo en una cantidad limitada de casos, y solo para mostrar información corta. 
Veamos una forma diferente de mostrar información. El mensaje se mostrara en una sudventana diferente a la del precio, y esta información se mostrara con objetos gráficos. Usar objetos gráficos tiene la gran ventaja que estos se pueden mover con la grafica, (a diferencia de la función Comment()) creando así una especie de línea de tiempo.

Para crear una subventana, hay que crear un indicador personalizado. El único propósito de este indicador es la creación de la sub-ventana, y no realizara ningún tipo de cálculo, ni tampoco dibujara líneas. El código de este indicador, al que vamos a llamar Inform.mq4 se muestra a continuación. 
//--------------------------------------------------------------------

// Inform.mq4

// Este código solo debe ser usado de manera educacional.

//--------------------------------------------------------------------

#property indicator_separate_window // Indicador separado de la ventana

//--------------------------------------------------------------------

int start() // función especial start()

{

}

//--------------------------------------------------------------------

En general, un programador puede colocar en el indicador el código que desee, y mejorar sus características. Por ejemplo usted puede mostrar unas líneas en una parte de la subventana. En el ejemplo anterior se muestra un código simple en que se crea una nueva subventana.

Función definida por el usuario Inform()

int Inform(int Mess_Number,int Number=0,double Value=0.0)
Esta función mostrara los objetos gráficos descritos más arriba en el indicador que vimos antes Inform.mq4. Esta función controla la posición de los objetos gráficos que se mostraran en la subventana: cada mensaje nuevo, se colocara en una línea más abajo del indicador y se coloreara con el color deseado, y las líneas anteriores se moverán a la parte superior del indicador. Si no hay actualizaciones en los datos en 15 segundos, los textos cambiaran su color a gris, para no crearle distracciones al usuario. 
Parámetros: 

Mess_Number - número del mensaje que puede tomar los siguientes valores:

· (cero) 0 - no se muestra ningún mensaje. Este valor se utilizar para reiniciar el contador del tiempo;
·  (menos uno) -1 - todos los objetos gráficos creados por la función serán suprimidos;
· (uno o mayor) – Si es un número mayor que cero, este número será tomado como el numero del mensaje que se mostrara en la subventana del indicador;

Number - número entero que se usara en algunos mensajes;

Value - número real que se usara en algunos mensajes;

La función Inform() que es la que crea los objetos gráficos, la vamos a colocar en un archivo separado, como en las otras funciones en un EA normal y luego la vamos a incluir en nuestro indicador. Este archivo se llamara Inform.mqh:

//----------------------------------------------------------------------------

// Inform.mqh

// Este código solo debe ser usado de manera educacional.

//----------------------------------------------------------------------- 1 --

// Función que muestra mensajes en objetos gráficos en la pantalla.

//----------------------------------------------------------------------- 2 --

int Inform(int Mess_Number, int Number=0, double Value=0.0)

{

// int Mess_Number // Numero del mensaje 
// int Number // Numero entero
// double Value // Numero real
int Win_ind; // Número de la ventana del indicador
string Graf_Text; // Texto del mensaje
color Color_GT; // Color del texto del mensaje
static int Time_Mess; // Ultima fecha de publicación del mensaje
static int Nom_Mess_Graf; // Numero de mensajes gráficos 
static string Name_Grf_Txt[30]; // Array con los textos de mensajes publicados
//----------------------------------------------------------------------- 3 --

Win_ind= WindowFind("inform"); // Buscando el numero de la ventana
if (Win_ind<0)return; // Si no hay tal ventana, salir
//----------------------------------------------------------------------- 4 --

if (Mess_Number==0) // Esto sucede en cada tick 
{

if (Time_Mess==0) return; // Si ya es gris sale
if (GetTickCount()-Time_Mess>15000)// El color no ha sido actualizado en el plazo de 15 seg
{

for(int i=0;i<=29; i++) // Líneas de color gris
ObjectSet( Name_Grf_Txt[i], OBJPROP_COLOR, Gray);

Time_Mess=0; // Bandera  que indica que todas las líneas son de color gris
WindowRedraw(); // Redibujo de los objetos

}

return; // Salimos de la función 
}

//----------------------------------------------------------------------- 5 --

if (Mess_Number==-1) // Esto se cumple en deinit()

{

for(i=0; i<=29; i++) // El índice de los objetos
ObjectDelete(Name_Grf_Txt[i]);// Eliminación de los objetos
return; // Salimos de la función
}

//----------------------------------------------------------------------- 6 --

Nom_Mess_Graf++; // Contador de mensajes 
Time_Mess=GetTickCount(); // La fecha del último mensaje 
Color_GT=Lime;

//----------------------------------------------------------------------- 7 --

switch(Mess_Number) // Elegir texto del mensaje según el numero 
{

case 1:

Graf_Text="Closed order Buy "+ Number;

PlaySound("Close_order.wav"); break;

case 2:

Graf_Text="Closed order Sell "+ Number;

PlaySound("Close_order.wav"); break;

case 3:

Graf_Text="Deleted pending order "+ Number;

PlaySound("Close_order.wav"); break;

case 4:

Graf_Text="Opened order Buy "+ Number;

PlaySound("Ok.wav"); break;

case 5:

Graf_Text="Opened order Sell "+ Number;

PlaySound("Ok.wav"); break;

case 6:

Graf_Text="Placed pending order "+ Number;

PlaySound("Ok.wav"); break;

case 7:

Graf_Text="Order "+Number+" modified into the market one";

PlaySound("Transform.wav"); break;

case 8:

Graf_Text="Reopened order "+ Number; break;

PlaySound("Bulk.wav");

case 9:

Graf_Text="Partly closed order "+ Number;

PlaySound("Close_order.wav"); break;

case 10:

Graf_Text="New minimum distance: "+ Number;

PlaySound("Inform.wav"); break;

case 11:

Graf_Text=" Not enough money for "+

DoubleToStr(Value,2) + " lots";

Color_GT=Red;

PlaySound("Oops.wav"); break;

case 12:

Graf_Text="Trying to close order "+ Number;

PlaySound("expert.wav"); break;

case 13:

if (Number>0)

Graf_Text="Trying to open order Sell..";

else

Graf_Text="Trying to open order Buy..";

PlaySound("expert.wav"); break;

case 14:

Graf_Text="Invalid password. EA doesn't function.";

Color_GT=Red;

PlaySound("Oops.wav"); break;

case 15:

switch(Number) // Elegir texto del mensaje según el numero del error
{

case 2: Graf_Text="Common error."; break;

case 129: Graf_Text="Wrong price. "; break;

case 135: Graf_Text="Price changed. "; break;

case 136: Graf_Text="No prices. Awaiting a new tick.."; break;

case 146: Graf_Text="Trading subsystem is busy"; break;

case 5 : Graf_Text="Old version of the terminal."; break;

case 64: Graf_Text="Account is blocked."; break;

case 133: Graf_Text="Trading is prohibited"; break;

default: Graf_Text="Occurred error " + Number;//otros errores
}

Color_GT=Red;

PlaySound("Error.wav"); break;

case 16:

Graf_Text="Expert Advisor works only for EURUSD";

Color_GT=Red;

PlaySound("Oops.wav"); break;

default:

Graf_Text="default "+ Mess_Number;

Color_GT=Red;

PlaySound("Bzrrr.wav");

}

//----------------------------------------------------------------------- 8 --

ObjectDelete(Name_Grf_Txt[29]); // Eliminación del objeto  29 en el array
for(i=29; i>=1; i--) // Ciclo en el índice del array...

{ // .. en los objetos gráficos 
Name_Grf_Txt[i]=Name_Grf_Txt[i-1];// Aumento de los objetos 
ObjectSet( Name_Grf_Txt[i], OBJPROP_YDISTANCE, 2+15*i);

}

Name_Grf_Txt[0]="Inform_"+Nom_Mess_Graf+"_"+Symbol(); // Object name

ObjectCreate (Name_Grf_Txt[0],OBJ_LABEL, Win_ind,0,0);// Creación 
ObjectSet (Name_Grf_Txt[0],OBJPROP_CORNER, 3 ); // Esquina
ObjectSet (Name_Grf_Txt[0],OBJPROP_XDISTANCE, 450);// Eje  Х

ObjectSet (Name_Grf_Txt[0],OBJPROP_YDISTANCE, 2); // Eje Y

// Текстовое описание объекта??
ObjectSetText(Name_Grf_Txt[0],Graf_Text,10,"Courier New",Color_GT);

WindowRedraw(); // Se redibujan todos los objetos. 
return;

}

//----------------------------------------------------------------------- 9 --

En el bloque 2 – 3 se declaran y describen las variables que se van a utilizar en la función.  Para guardar los nombres de los objetos, utilizaremos el array llamado Name_Grf_Txt. Según el parámetro colocado en la función, se crea un nuevo objeto grafico para mostrar un mensaje. La cantidad total de objetos es 30 y cada objeto representa una entrada de texto escrita en una línea. En caso de tener una resolución de pantalla grande, la cantidad máxima de objetos mostrados puede aumentar. 
En el bloque 3 – 4 se verifica que la ventana del indicador sea la correcta. Si no es la correcta, no se muestra ningún objeto grafico. Si no se ha agregado la subventana, el resto del EA funcionara correctamente, y las operaciones no serán afectadas. 

En el bloque 4 – 5, se analizan los colores de los menajes. Si a la función se le manda el parámetro Mess_Number=0 quiere decir que no hay mensajes. Si todos los objetos son grises ya, no sucede nada. Pero si el valor de Time_Mess es diferente a cero, el color de todos los objetos cambia a gris. 
Si (bloque 5 – 6) el parámetro de Mess_Number  es  -1, todos los objetos creados anteriormente se borran.  Esto puede ser necesario cuando se quita al EA de la grafica. En este caso, es común que cada programa que creo objetos gráficos, también los borre cuando se cierra el programa en la función especial deinit(). 

Si el programa llega hasta el bloque 6 – 7, significa que es necesario crear un nuevo objeto gráfico con las características requeridas y ponerlo en la parte más inferior del subwindow del indicador (“en la línea más baja”; aquí, el término “línea” es condicional. De hecho, la localización de objetos gráficos es determinada por las coordenadas que le demos). Cada objeto recién creado tiene un nombre único. Para darle el nombre a cada objeto, utilizamos el número histórico del mensaje, esta es la razón por la cual colocamos los nombres al revés en el bloque 6 – 7 (posteriormente el valor de la variable Nom_Mess_Graf se utilizara para formar un único nombre en bloque 8-9). Es acá donde se lleva la fecha de la última publicación de un mensaje y a los mensajes nuevos se le coloca el color verde. 

El bloque 7 – 8 consiste en un operador “switch” que depende del valor del mensaje que está en Mess_Number. Cada “caso” de este “switch”, representa un mensaje a mostrar, y este mensaje será copiado a la variable Graf_Text, mostrando así un mensaje apropiado según el caso.  En algunos casos también se cambia el color del texto, por ejemplo, rojo para los mensajes importantes. Todos los mensajes se acompañan con un sonido característico, que se reproducen con la función PlaySound () (ver archivos de sonido). LINK.
La creación de un nuevo objeto gráfico y el reemplazo de los existentes se realiza en el bloque 8-9. La cantidad máxima de objetos gráficos es limitada, así que un objeto (el más viejo) se elimina cada vez que se publica un nuevo mensaje. El resto de los objetos se mueve una línea hacia arriba. Los objetos se mueven cambiando la propiedad que representa las coordenadas horizontales. 

Después de que se han hecho todos los cálculos (todos los objetos se han movido una línea hacia arriba), un nuevo objeto con un nombre único y con las propiedades determinadas en el bloque 7- 8 se crea. El tipo de grafico es etiqueta de texto. Este tipo de permite al usuario mover el gráfico de precios de un lado a otro, sin que ello afecte a las posiciones de los mensajes.

La función Inform() puede ser llamada desde cualquier lugar del programa en que se necesite mostrar un mensaje de texto. Como resultado de un periodo largo de funcionamiento, los mensajes se acumularan en la ventana. El usuario puede ver el historial de los mensajes cuando se aumenta el tamaño de la subventana donde está el indicador. Opcionalmente se puede fijar el tamaño de la ventana para ver las líneas de texto que se necesiten ( se recomienda de 3 a 4 líneas) .

Es fácil ver que se puede agregar nuevos tipos de mensajes para mostrar. Solo con agregar un nuevo caso o “case” en el operador “switch” en el bloque 7-8.

